Maskinarkitektur og operativsystemer

NOTER
7.2.a.1



7.2.aPRIVATE 


Implementering af procedurer og funktioner

Bjørk Busch
I assembler er der i forbindelse med procedurer og funktioner flere måder at overføre parametre og returværdier på.

Den ene måde er ved at anvende registrene til indhold eller som pointere, en anden måde er ved at anvende stakken.

Når man anvender stakken til parameter overførsel kan dette igen gøres på flere måder.

PASCAL anvender stakken på en måde og sproget C gør det en anden. Hovedforskellen er dog primært om det er det kaldende program eller underrutinen, der skal ryde op på stakken.

I PASCAL er det underrutinens opgave at ryde op på stakken og i C er det kaldende program, der selv ryder op.

I det efterfølgende vises hvordan stakken kan anvendes efter den metode, som PASCAL anvender.

Parameteroverførsel ved brug af stakken.


             STAK før procedurekald


Low adr.   │                        │



           │                        │



           │                        │



STAKTOP    │                        │



SS:[SP] ‑> ├────────────────────────┤



           │xxxxxxxxxxxxxxxxxxxxxxxx│



           │x    brugt før kald    x│



           │xxxxxxxxxxxxxxxxxxxxxxxx│



High adr.  └────────────────────────┘

Procedurekald med value parametre.
Proceduren kan være defineret som:


PROCEDURE minproc (x,y:integer)
Opstart af proceduren:
        PUSH    xmain
        PUSH    ymain
        CALL    minproc


          STAK ved start af procedure


Low adr.   │                        │              



           │                        │              



STAKTOP    │                        │              



SS:[SP] ‑> ├────────────────────────┤ 



           │  Returadresse          │              



ACTIVATION-├────────────────────────┤ 



RECORD     │ Kopi af parameter 2 (Y)│ 



           ├────────────────────────┤              



           │ Kopi af parameter 1 (X)│              



-----------├────────────────────────┤ 



           │xxxxxxxxxxxxxxxxxxxxxxxx│              



           │x    brugt før kald    x│              



           │xxxxxxxxxxxxxxxxxxxxxxxx│              



High adr.  └────────────────────────┘              

Indledning i procedure (PROLOG):
"FASFRYS" udganspunkt returadr. og parametre

        PUSH    BP
        MOV     BP,SP


            STAK efter "fastfrysning"


Low adr.   │                        │              



           │                        │              



           │                        │              



           │                        │              



           │                        │              



STAKTOP    │                        │              



SS:[SP] ‑> ├────────────────────────┤ <= SS:[BP]   


           │  BP (gemt)             │              



-----------├────────────────────────┤ <‑ SS:[BP+02]



           │  Returadresse          │              



ACTIVATION-├────────────────────────┤ <‑ SS:[BP+04]



RECORD     │ Kopi af parameter 2 (Y)│ (hvis 2 byte)



           ├────────────────────────┤ <‑ SS:[BP+06]



           │ Kopi af parameter 1 (X)│ (hvis 2 byte)



-----------├────────────────────────┤ <‑ SS:[BP+08]



           │xxxxxxxxxxxxxxxxxxxxxxxx│



           │x    brugt før kald    x│



           │xxxxxxxxxxxxxxxxxxxxxxxx│



High adr.  └────────────────────────┘

Indledning i procedure fortsat ‑ hvis lokale variable:
Der skaffes hukommelse til lokale variable (her 2 integer)

        SUB     Sp,4


      STAK efter plads til lokale variable


Low adr.   │                        │              



           │                        │              



STAKTOP    │                        │              



SS:[SP] ‑> ├────────────────────────┤ <‑ SS:[BP‑04]



           │  Lokal variabel 2      │ (hvis 2 byte)



           ├────────────────────────┤ <‑ SS:[BP‑02]



           │  Lokal variabel 1      │ (hvis 2 byte)



           ├────────────────────────┤ <=== SS:[BP]


           │  BP (gemt)             │              



-----------├────────────────────────┤ <‑ SS:[BP+02]



           │  Returadresse          │              



ACTIVATION-├────────────────────────┤ <‑ SS:[BP+04]



RECORD     │ Kopi af parameter 2 (Y)│ 



           ├────────────────────────┤ <‑ SS:[BP+06]



           │ Kopi af parameter 1 (X)│ 



-----------├────────────────────────┤ <‑ SS:[BP+08]



           │xxxxxxxxxxxxxxxxxxxxxxxx│



           │x    brugt før kald    x│



           │xxxxxxxxxxxxxxxxxxxxxxxx│



High adr.  └────────────────────────┘
Senere i procedure hvor Arbejdsregistrene AX og BX gemmes:

        PUSH    AX
        PUSH    BX


      STAK efter save af AX og BX for brug


Low adr.   │                        │



           │                        │



STAKTOP    │                        │



SS:[SP] ‑> ├────────────────────────┤



           │  BX (gemt)             │



           ├────────────────────────┤



           │  AX (gemt)             │



           ├────────────────────────┤ <‑ SS:[BP‑04]



           │  Lokal variabel 2      │



           ├────────────────────────┤ <‑ SS:[BP‑02]



           │  Lokal variabel 1      │



           ├────────────────────────┤ <=== SS:[BP]


           │  BP (gemt)             │



-----------├────────────────────────┤ <‑ SS:[BP+02]



           │  Returadresse          │



ACTIVATION-├────────────────────────┤ <‑ SS:[BP+04]



RECORD     │ Kopi af parameter 2 (Y)│



           ├────────────────────────┤ <‑ SS:[BP+06]



           │ Kopi af parameter 1 (X)│



-----------├────────────────────────┤ <‑ SS:[BP+08]



           │xxxxxxxxxxxxxxxxxxxxxxxx│



           │x    brugt før kald    x│



           │xxxxxxxxxxxxxxxxxxxxxxxx│



High adr.  └────────────────────────┘

Før afslutning af procedure hvor Arbejdsregistrene

AX og BX bliver reetableret:

        POP     BX
        POP     AX


        STAK efter restore af AX og BX


Low adr.   │                        │



           │                        │



STAKTOP    │                        │



SS:[SP] ‑> ├────────────────────────┤ <‑ SS:[BP‑04]



           │  Lokal variabel 2      │



           ├────────────────────────┤ <‑ SS:[BP‑02]



           │  Lokal variabel 1      │



           ├────────────────────────┤ <=== SS:[BP]


           │  BP (gemt)             │



-----------├────────────────────────┤ <‑ SS:[BP+02]



           │  Returadresse          │



ACTIVATION-├────────────────────────┤ <‑ SS:[BP+04]



RECORD     │ Kopi af parameter 2 (Y)│



           ├────────────────────────┤ <‑ SS:[BP+06]



           │ Kopi af parameter 1 (X)│



-----------├────────────────────────┤ <‑ SS:[BP+08]



           │xxxxxxxxxxxxxxxxxxxxxxxx│



           │x    brugt før kald    x│



           │xxxxxxxxxxxxxxxxxxxxxxxx│



High adr.  └────────────────────────┘
Afslutning af procedure (EPILOG)
        MOV     SP,BP
        POP     BP


      STAK efter afslutning af procedure


Low adr.   │                        │



           │                        │



           │                        │



           │                        │



           │                        │



           │                        │



           │                        │



STAKTOP    │                        │



SS:[SP] ‑> ├────────────────────────┤



           │  Returadresse          │



ACTIVATION-├────────────────────────┤



RECORD     │ Kopi af parameter 2 (Y)│



           ├────────────────────────┤



           │ Kopi af parameter 1 (X)│



-----------├────────────────────────┤



           │xxxxxxxxxxxxxxxxxxxxxxxx│



           │x    brugt før kald    x│



           │xxxxxxxxxxxxxxxxxxxxxxxx│



High adr.  └────────────────────────┘

Returnering fra procedure (parametre fyldte 4 bytes (2 word))

        RET     4


;returner og ryd op på stak

      STAK efter returnering fra procedure


Low adr.   │                        │



           │                        │



           │                        │



           │                        │



           │                        │



           │                        │



           │                        │



SSTAKTOP   │                        │



SS:[SP] ‑> ├────────────────────────┤



           │xxxxxxxxxxxxxxxxxxxxxxxx│



           │x    brugt før kald    x│



           │xxxxxxxxxxxxxxxxxxxxxxxx│



High adr.  └────────────────────────┘
Procedurer med returparametre.
I forbindelse med procedurer med returparametre overføres adressen på parametrene på stakken istedet for indholdet.

Procedureindledningen er som tidligere beskrevet, og stakken ser efter denne således ud for en procedure med to VAR-parametre og een lokal variabel af typen integer/word.

Proceduren kan være defineret som:


PROCEDURE PROC2 (VAR X,Y: INTEGER)
Opstart af proceduren:
        LEA     Ax,xmain
        PUSH    Ds
        PUSH    Ax
        LEA     Ax,ymain
        PUSH    Ds
        PUSH    Ax
        CALL    PROC2
Indledning i procedure (PROLOG):
"FASFRYS" udganspunkt, returadr., parametre og afsæt plads til lokalvariabel.

        PUSH    BP
        MOV     BP,SP
        SUB     Sp,2


      STAK efter plads til lokale variable


Low adr.   │                         │



           │                         │



STAKTOP    │                         │



SS:[SP] ‑> ├─────────────────────────┤ <‑ SS:[BP‑02]



           │  Lokal variabel 1 (A)   │



           ├─────────────────────────┤ <=== SS:[BP]


           │  BP (gemt)              │



-----------├─────────────────────────┤ <‑ SS:[BP+02]



           │  Returadresse           │



ACTIVATION-├─────────────────────────┤ <‑ SS:[BP+04]



RECORD     │ Ofs.adr på parameter 2  │



           ├─────────────────────────┤ <‑ SS:[BP+06]



           │ Seg.adr på parameter 2  │



           ├─────────────────────────┤ <‑ SS:[BP+08]



           │ Ofs.adr på parameter 1  │



           ├─────────────────────────┤ <‑ SS:[BP+10]



           │ Seg.adr på parameter 1  │



-----------├─────────────────────────┤ <‑ SS:[BP+12]



           │xxxxxxxxxxxxxxxxxxxxxxxxx│



           │x    brugt før kald     x│



           │xxxxxxxxxxxxxxxxxxxxxxxxx│



High adr.  └─────────────────────────┘

Hvis proceduren skal ombytte X og Y på følgende måde:
        A := X;

        X := Y;

        Y := A;

Det centrale i proceduren kan nu ske på følgende måde:
        LES    Bx,Ss:[Bp+8]        ; Bx:=ofs. og Es:=seg. på X

        MOV    Ax,Es:[Bx]          ; Ax := X

        MOV    Ss:[Bp-2],Ax        ; A  := Ax

        LES    Bx,Ss:[Bp+4]        ; Es:Bx := adr. på Y

        MOV    Ax,Es:[Bx]          ; Ax := Y

        LES    Bx,Ss:[Bp+8]        ; Es:Bx := adr. på X

        MOV    Es:[Bx],Ax          ; X := Ax

        MOV    Ax,[Bp-2]           ; Ax := A

        LES    Bx,Ss:[Bp+4]        ; Es:Bx := adr. på Y

        MOV    Es:[Bx],Ax          ; Y := Ax

Bemærk at LES-instruktionen kan erstatte to MOV-instruktioner

Som det sikker fremgår, er det muligt at spare nogle instruktioner, hvis man bruger flere arbejdsregistre.

Afslutning af procedure (EPILOG)
        MOV     SP,BP
        POP     BP
Returnering fra procedure (adresse-parametre fyldte 8 bytes (hver adresse var på 2*2 word))

        RET     8


;returner og ryd op på stak

Eksempel på hvordan procedure GETDATE kan se ud.
Proceduren er i pascal defineret på følgende måde:

Procedure GetDate (var aa:integer; var mm, dd, ugedg: byte);
GetDate Proc    Near
        PUSH    Bp              ;gem basispointer

        MOV     Bp,Sp           ;Bp ‑> staktop

                                ;   sidste parameter er 4 højere

        PUSH    Ax              ;gem arbejdsregistre

        PUSH    Bx

        PUSH    Cx

        PUSH    Dx

        PUSH    Es

        MOV     Ah,2Ah          ;funktion GET DATE

        INT     21h             ;Dos service funktion

        LES     Bx,Ss:[Bp+4]    ;Es:Bx := adr. på sidste parm

        MOV     Es:[Bx],Al      ;returner ugedag

        LES     Bx,Ss:[Bp+8]    ;Es:Bx := adr. på 2. sidste parm

        MOV     Es:[Bx],Dl      ;returner dag

        LES     Bx,Ss:[Bp+12]   ;Es:Bx := adr. på 3. sidste parm

        MOV     Es:[Bx],Dh      ;returner måned

        LES     Bx,Ss:[Bp+16]   ;Es:Bx := adr. på 4. sidste parm

        MOV     Es:[Bx],Cx      ;returner årstal

        POP     Es              ;restore arbejdsregistre

        POP     Dx

        POP     Cx

        POP     Bx

        POP     Ax

        POP     Bp              ; restore basispointer

        RET     16              ;returner og juster stak (parm 4*(2 ord))

GetDate EndP

Eksempel på brug af proceduren:
Variable i datasegmentet, som antages at være sat op i DS.

.......................................

AARSTAL DW      ?

MAANED  DB      ?

DAG     DB      ?

UGEDAG  DB      ?

.......................................

        LEA     AX,AARSTAL

        PUSH    DS               ; 1. parameters adresse

        PUSH    AX               ;     lægges på stak

        LEA     AX,MAANED

        PUSH    DS               ; 2. parameters adresse

        PUSH    AX               ;     lægges på stak

        LEA     AX,DAG

        PUSH    DS               ; 3. parameters adresse

        PUSH    AX               ;     lægges på stak

        LEA     AX,UGEDAG

        PUSH    DS               ; 4. parameters adresse

        PUSH    AX               ;     lægges på stak

        CALL    GetDate

.......................................

Funktioner:
I forbindelse med Funktioner returneres funktionsværdien på stakken.

Det kaldende program afsætter plads til funktionsværdien.
Bliver funktionen kaldt med parametre, overføres disse på samme måde som der gælder for procedurer, idet der først sættes plads til funktionsværdien og herefter udlægges parametre.

Funktionen kan være defineret som:


FUNCTION SUM (X, Y: INTEGER) : INTEGER
      STAK efter plads til lokale variable
Low adr.   │                         │

STAKTOP    │                         │

SS:[SP] ‑> ├─────────────────────────┤ <‑ SS:[BP‑02]

           │  Lokal variabel 1 (A)   │

           ├─────────────────────────┤ <=== SS:[BP]
           │  BP (gemt)              │

-----------├─────────────────────────┤ <‑ SS:[BP+02]

           │  Returadresse           │

ACTIVATION-├─────────────────────────┤ <‑ SS:[BP+04]

RECORD     │ Kopi af parameter 2 (Y) │

           ├─────────────────────────┤ <‑ SS:[BP+06]

           │ Kopi af parameter 1 (X) │

           ├─────────────────────────┤ <‑ SS:[BP+08]

           │ Funktionsværdi (indhold)│

-----------├─────────────────────────┤ <‑ SS:[BP+10]

           │xxxxxxxxxxxxxxxxxxxxxxxxx│

           │x    brugt før kald     x│

           │xxxxxxxxxxxxxxxxxxxxxxxxx│

High adr.  └─────────────────────────┘

Hvis funktionen skal returnere summen af de 2 parametre kan den se ud på følgende måde:

SUM     Proc    Near
        PUSH    Bp              ; gem basispointer

        MOV     Bp,Sp           ; Bp ‑> staktop

                                ;    sidste parameter er 4 højere

        PUSH    Ax              ; gem arbejdsregistre

        MOV     Ax,SS:[Bp+6]    ; Ax := X
        ADD     Ax,SS:[Bp+4]    ; Ax := Ax + Y
        MOV     SS:[Bp+8],Ax    ; funktionsværdi := Ax
        POP     Ax              ; restore arbejdsregister

        POP     Bp              ; restore basispointer

        RET     4               ; returner og juster stak, idet

                                ;     funktionsværdi efterlades på stak-top

SUM     EndP
Eksempel på brug af funktionen:
Variable i datasegmentet, som antages at være sat op i DS.

.......................................

TAL1    DW      3

TAL2    DW      4

SUMTAL  DW      ?

.......................................

        SUB     SP,2             ; Afsæt plads til funktionsværdi

        PUSH    TAL1             ; overfør parameter 1 til stak (kopi)

        PUSH    TAL2             ; overfør parameter 2 til stak (kopi)

        CALL    SUM

        POP     SUMTAL           ; SUMTAL := funktionsværdi

.......................................

Bemærk at PUSH og POP kun kan anvendes ved word-parametre, ellers må man selv justere SP-register og bruge MOV-instruktionen til at flytte data med.
Flere eksempler på procedurer og funktioner.
PROCEDURE READCHAR (VAR TEGN: CHAR);
Hent tegn fra tastaturbuffer.

Procedure kodet i assembler.
ReadChar Proc    Near

         PUSH    Bp              ;gem basispointer

         MOV     Bp,Sp           ;Bp ‑> staktop

                                 ;   sidste parameter er 4 højere

         PUSH    Ax              ;gem arbejdsregistre

         PUSH    Bx

         PUSH    Es

         MOV     Ah,08h          ;funktion INPUT ‑ ECHO + CTRL C

         INT     21h             ;Dos service funktion

         LES     Bx,Ss:[Bp+4]    ;Es:Bx := adr. på sidste parm

         MOV     Es:[Bx],Al      ;returner tegn

         POP     Es              ;restore arbejdsregistre

         POP     Bx

         POP     Ax

         POP     Bp              ; restore basispointer

         RET     4               ;returner og juster stak

ReadChar EndP

Anvendelse fra assembler.
TEGN     DB      ?

.......................................

        LEA     AX,TEGN

        PUSH    DS               ; 1. parameters adresse

        PUSH    AX               ;     lægges på stak

        CALL    ReadChar

Samme rutine men som funktion:
FUNCTION READKEY : CHAR;
Procedure kodet i assembler.
ReadKey  Proc    Near

         PUSH    Bp              ;gem basispointer

         MOV     Bp,Sp           ;Bp ‑> staktop

                                 ;   sidste parameter er 4 højere

         PUSH    Ax              ;gem arbejdsregistre

         MOV     Ah,08h          ;funktion INPUT ‑ ECHO + CTRL C

         INT     21h             ;Dos service funktion

         MOV     Ss:[Bp+4],Al    ;Returner funktionsværdi     

         POP     Ax

         POP     Bp              ; restore basispointer

         RET     0               ; returner og juster stak

ReadKey EndP

Anvendelse fra assembler.
TEGN     DB      ?

.......................................

        SUB     Sp,1             ; Sæt plads til funktionsværdi

        CALL    ReadKey

        MOV     Bp,Sp            ; Hent funktionsværdi

        MOV     Al,Ss:[Bp]       ;   fra stakken

        ADD     Sp,1             ; Fjern fra stak

        MOV     TEGN,Al          ; Overfør funktionsværdi

Som det kan ses er det noget besværlig når der arbejdes med ulige antal bytes på stak. Der kan anvendes word istedet, hvor den ene byte så blot ignoreres.
PROCEDURE WRITECHR (TEGN: CHAR);
Skriv tegn på skærm.

procedure kodet i assembler.
WriteChr Proc    Near

         PUSH    Bp              ;gem basispointer

         MOV     Bp,Sp           ;Bp ‑> staktop

                                 ;   sidste parameter er 4 højere

         PUSH    Ax              ;gem arbejdsregistre

         PUSH    Dx

         MOV     Dl,Ss:[Bp+4]    ;Dl := parameterværdi

         MOV     Ah,02h          ;funktion OUTPUT CHARACTER

         INT     21h             ;Dos service funktion

         POP     Dx              ;restore arbejdsregistre

         POP     Ax

         POP     Bp              ;restore basispointer

         RET     1               ;returner og juster stak

WriteChr EndP

Opstart fra assembler.
TEGN     DB      ?

.......................................

        SUB     Sp,1             ; Plads til parameter

        MOV     Bp,Sp            ; Overfør parameter

        MOV     Al,Tegn          ;   til stak

        MOV     Ss:[Bp],Al       ;(ulige antal byte er besværlig)

        CALL    WriteChr

Som det kan ses er det noget besværlig når der arbejdes med ulige antal bytes på stak. Der kan anvendes word istedet, hvor den ene byte så blot ignoreres.

Hvis der anvedes word istedet ville procedure og kald se således ud:
WriteChr Proc    Near

         PUSH    Bp              ;gem basispointer

         MOV     Bp,Sp           ;Bp ‑> staktop

                                 ;   sidste parameter er 4 højere

         PUSH    Ax              ;gem arbejdsregistre

         PUSH    Dx

         MOV     Dl,SS:[Bp+4]    ;Dl := parameterværdi

         MOV     Ah,02h          ;funktion OUTPUT CHARACTER

         INT     21h             ;Dos service funktion

         POP     Dx              ;restore arbejdsregistre

         POP     Ax

         POP     Bp              ;restore basispointer

         RET     2               ;returner og juster stak

WriteChr EndP

Opstart fra assembler.
TEGN     DB      ?

.......................................

        PUSH    TEGN             ;Overfør parameter til stak

        CALL    WriteChr

Det blev det noget enklere af ikke !!!
Bemærk at proceduren skal tage 1 mere af stakken.
I praksis anvendes normalt 2 byte, da det er enklere.
FUNCTION GETCHARXY (X, Y: Integer) : CHAR;
Hent tegn fra skærmbuffer position X,Y (hjørne = 1,1).

Procedure kodet i assembler.
SKRMSEG    EQU     0B800h          ;Kunne også findes via. skærmmode

GetCharXY  Proc    Near

           PUSH    Bp              ;gem basispointer

           MOV     Bp,Sp           ;Bp ‑> staktop

                                   ;   sidste parameter er 4 højere

           PUSH    Ax              ;gem arbejdsregistre

           PUSH    Si

           PUSH    Es

;






;indexadresse=(((Y-1)*80)+X-1)*2

           MOV     Ax,Ss:[Bp+4]    ;Ax := Y

           DEC     Ax              ;ryk start til 0

           MOV     Si,80           ;tegn pr. linie

           MUL     Si              ;Si := start tegn for linie

           ADD     Ax,Ss:[Bp+6]    ;adder kolonne = X

           DEC     Ax              ;ryk X-start til 0

           SHL     Ax,1            ;Ax := tegnoffset 2 byte pr. tegn

           MOV     Si,Ax           ;Si := udvalgt tegnadresse

           MOV     Ax,SKRMSEG      
           MOV     Es,Ax           ;opsæt skærmbuffer segmentadr.

           MOV     Al,Es:[Si]      ;hent ascii fra skærmbuffer

           XOR     Ah,Ah           ;udvid til word - nulstil Ah

           MOV     Ss:[Bp+8],Ax    ;overfør funktionsværdi til stak

           POP     Es

           POP     Si

           POP     Ax

           POP     Bp              ;restore basispointer

           RET     4               ;returner og juster stak

                                   ;funktionsværdi bliver på stak

GetCharXY  EndP

Opstart fra assembler.
TEGN     DB      ?

X        DW      4

Y        DW      12

.......................................

        SUB     Sp,2             ;plads til funktionsværdi

        PUSH    X                ;1. parameters indhold på stak

        PUSH    Y                ;2. parameters indhold på stak

        CALL    GetCharXY

        POP     Ax               ;Hent funktionsværdi og

        MOV     Tegn,Al          ;overfør den til Tegn.

Det skal bemærkes, at der her er anvendt word parametre i procedure og "afkortning" først sker efter kaldet ved overførselen af funktionsværdien til TEGN.
Det er muligt at gøre procedurerne mere "læsevenlige" ved brug af EQU sætninger. Dette vises på næste side med samme procedure.
FUNCTION GETCHARXY (X, Y: Integer) : CHAR;
Denne gang mere "læsevenlig" ved brug af EQU-sætninger.
SKRMSEG    EQU     0B800h
Y          EQU     WORD PTR Ss:[Bp+4]    ; 2. parameter

X          EQU     WORD PTR Ss:[Bp+6]    ; 1. parameter

FUNK       EQU     WORD PTR Ss:[Bp+8]    ; funktionsværdi - retur

GetCharXY  Proc    Near

           PUSH    Bp              ;gem basispointer

           MOV     Bp,Sp           ;Bp ‑> staktop

                                   ;   sidste parameter er 4 højere

           PUSH    Ax              ;gem arbejdsregistre

           PUSH    Di

           PUSH    Es

           MOV     Ax,Y            ;Ax := Y

           DEC     Ax              ;ryk start til 0

           MOV     Di,80           ;tegn pr. linie

           MUL     Di              ;Di := start tegn for linie

           ADD     Ax,X            ;adder X

           DEC     Ax              ;ryk start til 0

           SHL     Ax,1            ;Ax := tegnoffset 2 byte pr. tegn

           MOV     Di,Ax           ;Di := udvalgt tegnadresse

           MOV     Ax,SKRMSEG      
           MOV     Es,Ax           ;opsæt skærmbuffer segmentadr.

           MOV     Al,Es:[Di]      ;hent ascii fra skærmbuffer

           XOR     Ah,Ah           
           MOV     FUNK,Ax         ;overfør funktionsværdi til stak

           POP     Es

           POP     Di

           POP     Ax

           POP     Bp              ;restore basispointer

           RET     4               ;returner og juster stak

                                   ;funktionsværdi bliver på stak

GetCharXY  EndP

Under oversættelsen erstattes symbolnavne fra EQU med det de er defineret som og der er derfor ingen forskel på den oversatte procedure i forhold til foregående udgave.
Prøv selv at lave følgende procedurer/funktioner med tilhørende eksempel på kald:
Du kan også prøve at afteste under CODEVIEW.

Til løsning kan anvendes BIOS-kald 10h.

1)
PROCEDURE GOTOXY ( X, Y: Integer )


Placer cursor

2)
FUNCTION GETX : Integer;   og   FUNCTION GETY : Integer;


Returner cursor position, henholdsvis X og Y koordinat.

3)
PROCEDURE GETXY (VAR X, Y: Integer);


Returner både X og Y koordinat

HER SLUTTER DEN EGENTLIGE NOTE !!!

HELE EKSEMPLER FØLGER - KAN FLYTTES TIL DOSFIL-FORMAT
Code     Segment

         Assume Cs:Code,Ds:Code,Ss:Code

ReadChar Proc    Near

         PUSH    Bp              ;gem basispointer

         MOV     Bp,Sp           ;Bp ‑> staktop

                                 ;   sidste parameter er 4 højere

         PUSH    Ax              ;gem arbejdsregistre

         PUSH    Bx

         PUSH    Es

         MOV     Ah,08h          ;funktion INPUT ‑ ECHO + CTRL C

         INT     21h             ;Dos service funktion

         LES     Bx,SS:[Bp+4]    ;Es:Bx := adr. på sidste parm

         MOV     Es:[Bx],Al      ;returner tegn

         POP     Es              ;restore arbejdsregistre

         POP     Bx

         POP     Ax

         POP     Bp              ; restore basispointer

         RET     4               ;returner og juster stak

ReadChar EndP

;Eksempel på brug af proceduren:

;Variable i datasegmentet, som antages at være sat op i DS.

;.......................................

TEGN     DB      ?

;.......................................

Start:

        MOV     Ax,Seg Code

        MOV     Ds,Ax

        LEA     AX,TEGN

        PUSH    DS               ; 1. parameters adresse

        PUSH    AX               ;     lægges på stak

        CALL    ReadChar

;.......................................

code    Ends

        End     START

Code     Segment

         Assume Cs:Code,Ds:Code,Ss:Code

;Eksempel på brug af proceduren:

;Variable i datasegmentet, som antages at være sat op i DS.

;.......................................

TEGN     DB      ?

;.......................................

Start:

        MOV     Ax,Seg Code

        MOV     Ds,Ax

        SUB     Sp,1             ; Sæt plads til funktionsværdi

        CALL    ReadKey

        MOV     Bp,Sp            ; Hent funktionsværdi

        MOV     Al,Ss:[Bp]       ;   fra stakken

        ADD     Sp,1             ; Fjern fra stak

        MOV     TEGN,Al          ; Overfør funktionsværdi

        MOV     AX,4C00h

        INT     21h

ReadKey  Proc    Near

         PUSH    Bp              ;gem basispointer

         MOV     Bp,Sp           ;Bp ‑> staktop

                                 ;   sidste parameter er 4 højere

         PUSH    Ax              ;gem arbejdsregistre

         MOV     Ah,08h          ;funktion INPUT ‑ ECHO + CTRL C

         INT     21h             ;Dos service funktion

         MOV     SS:[Bp+4],Al    ;Returner funktionsværdi     

         POP     Ax

         POP     Bp              ; restore basispointer

         RET     0               ; returner og juster stak

ReadKey EndP

;.......................................

code    Ends

        End     START

Code     Segment

         Assume Cs:Code,Ds:Code,Ss:Code

WriteChr Proc    Near

         PUSH    Bp              ;gem basispointer

         MOV     Bp,Sp           ;Bp ‑> staktop

                                 ;   sidste parameter er 4 højere

         PUSH    Ax              ;gem arbejdsregistre

         PUSH    Dx

         MOV     Dl,SS:[Bp+4]    ;Dl := parameterværdi

         MOV     Ah,02h          ;funktion OUTPUT CHARACTER

         INT     21h             ;Dos service funktion

         POP     Dx              ;restore arbejdsregistre

         POP     Ax

         POP     Bp              ;restore basispointer

         RET     1               ;returner og juster stak

WriteChr EndP

;Eksempel på brug af proceduren:

;Variable i datasegmentet, som antages at være sat op i DS.

;.......................................

TEGN     DB      '&'

;.......................................

Start:

        MOV     Ax,Seg Code

        MOV     Ds,Ax

        SUB     Sp,1             ; Plads til parameter

        MOV     Bp,Sp            ; Overfør parameter

        MOV     Al,Tegn          ;   til stak

        MOV     Ss:[Bp],Al       ;(ulige antal byte er besværlig)

        CALL    WriteChr

        MOV     Ax,4C00h

        INT     21h

;.......................................

code    Ends

        End     START

